

Mini-course on GAP – Lecture 1

Leandro Vendramin

Universidad de Buenos Aires

Dalhousie University, Halifax
January 2020



GAP is a system for **computational discrete algebra**. It is freely available here: <http://www.gap-system.org/>

What are we going to do here?

Outline:

- ▶ Arithmetics
- ▶ Basic programming
- ▶ Linear algebra
- ▶ Elementary group theory
- ▶ Advanced group theory, representation theory
- ▶ Advanced applications such as combinatorial knot theory and non-commutative algebras

Immediately after running GAP we will see some information related to the distribution we have installed. We will also see that GAP is ready:

```
gap>
```

To close GAP one uses `quit`:

```
gap> quit;
```

Every command should end with the symbol `;` (semicolon). The symbol `;;` (double semicolon) also is used to end a command but it means that no screen output will be produced.

```
gap> 2+5;;
```

```
gap> 2+5;
```

```
7
```

To see information related commands or functions, tutorials and manuals one uses the symbol ? (question mark). Here we have some examples:

```
gap> ?tutorial
gap> ?sets
gap> ?help
gap> ?permutations
gap> ?Eigenvalues
gap> ?CyclicGroup
gap> ?FreeGroup
gap> ?SylowSubgroup
```

To make the command line more readable one could use the symbol `\` (backslash):

```
gap> # Let us compute 1+2+3
gap> 1\
> +2\
> +3;
6
```

The function `LogTo` saves the subsequent interaction to be logged to a file. It means that everything you see on your terminal will also appear in this file.

```
gap> # Save the output to the file mylog  
gap> LogTo("mylog");
```

This is **extremely useful**! When the function `LogTo` is called with no parameters GAP will stop writing a log file.

```
gap> # Stop saving the output  
gap> LogTo();
```

One can do **basic arithmetic** operations with rational numbers:

```
gap> 1+1;
```

```
2
```

```
gap> 2*3;
```

```
6
```

```
gap> 8/2;
```

```
4
```

```
gap> (1/3)+(2/5);
```

```
11/15
```

```
gap> 2*(-6)+4;
```

```
-8
```

```
gap> NumeratorRat(3/5);
```

```
3
```

```
gap> DenominatorRat(3/5);
```

```
5
```

One uses `mod` to obtain the remainder after division of a by m , where a is the dividend and m is the divisor.

```
gap> 6 mod 4;
```

```
2
```

```
gap> -6 mod 5;
```

```
4
```

There are several functions that one can use for specific purposes. For example `Factors` returns the **factorization of an integer** and `IsPrime` detects whether an integer is **prime** or not.

```
gap> Factors(10);  
[ 2, 5 ]  
gap> Factors(18);  
[ 2, 3, 3 ]  
gap> IsPrime(1800);  
false  
gap> Factors(37);  
[ 37 ]  
gap> IsPrime(37);  
true
```

Other useful functions: Sqrt computes **square roots**, Factorial computes the **factorial** of a positive integer, Gcd computes the **greatest common divisor** of a finite list of integers, Lcm computes the **least common multiple**.

```
gap> Sqrt(25);
```

```
5
```

```
gap> Factorial(15);
```

```
1307674368000
```

```
gap> Gcd(10,4);
```

```
2
```

```
gap> Lcm(10,4,2,6);
```

```
60
```

We can also work in **cyclotomic fields**. CF creates a cyclotomic field. To create **primitive roots** of 1 one uses the function E. More precisely: $E(n)$ returns $e^{2\pi i/n}$.

```
gap> E(6) in Rationals;  
false  
gap> E(6) in Cyclotomics;  
true  
gap> E(3) in CF(3);  
true  
gap> E(3) in CF(4);  
false  
gap> E(3)^2+E(3);  
-1  
gap> E(6);  
-E(3)^2
```

Typically, cyclotomic numbers will be represented as rational linear combinations of primitive roots of 1.

Inverse (resp. AdditiveInverse) returns the multiplicative (resp. additive) inverse of an element.

```
gap> AdditiveInverse(2/3);  
-2/3  
gap> Inverse(2/3);  
3/2  
gap> AdditiveInverse(E(7));  
-E(7)  
gap> Inverse(E(7));  
E(7)^6
```

Exercise: Conway FRACTRAN language

FRACTRAN is a programming language invented by J. Conway. A FRACTRAN program is simply an ordered list of positive rationals together with an initial positive integer input n . The program is run by updating the integer n as follows:

- ▶ For the first rational f in the list for which $nf \in \mathbb{Z}$, replace n by nf .
- ▶ Repeat this rule until no rational in the list produces an integer when multiplied by n , then stop.

Write an implementation of the FRACTRAN language.

Starting with $n = 2$, the program

$$\frac{17}{65}, \frac{133}{34}, \frac{17}{19}, \frac{23}{17}, \frac{2233}{69}, \frac{23}{29}, \frac{31}{23}, \frac{74}{341}, \frac{31}{37}, \frac{41}{31}, \frac{129}{287}, \frac{41}{43}, \frac{13}{41}, \frac{1}{13}, \frac{1}{3}$$

produces the sequence

$$2, 15, 825, 725, 1925, 2275, 425, 390, 330, 290, 770 \dots$$

In 1987, J. Conway proved that this sequence contains the set

$$\{2^p : p \text{ prime}\}.$$

See <https://oeis.org/A007542> for more information.

Exercise: Conway “look and say” sequence

The first terms of Conway’s “look and say” sequence are

1
11
21
1211
111221
312211

After guessing how each term is computed, write a script to create the first terms of the sequence.

To solve these two exercises we need some basic programming: conditionals, functions, strings, lists, ranges, sets, records, loops. . .
We will see these things tomorrow!

Let us review some basic mathematical objects in GAP :

- ▶ Permutations
- ▶ Finite fields
- ▶ Matrices

Permutations

A **permutation** in n letters is a bijective map

$$\sigma: \{1, \dots, n\} \rightarrow \{1, \dots, n\}.$$

For example, the permutation $\begin{pmatrix} 1234 \\ 3124 \end{pmatrix}$ is the bijective map such that $1 \mapsto 3$, $2 \mapsto 1$, $3 \mapsto 2$ and $4 \mapsto 4$.

Usually one writes a permutation as a product of disjoint cycles. For example:

$$\begin{pmatrix} 1234 \\ 2413 \end{pmatrix} = (1243), \quad \begin{pmatrix} 12345 \\ 21435 \end{pmatrix} = (12)(34)(5) = (12)(34).$$

The permutation $\begin{pmatrix} 12345 \\ 21435 \end{pmatrix} = (12)(34)$ in GAP is $(1,2)(3,4)$.

Permutations

The function `IsPerm` checks whether some object is a permutation.
Let us see some examples:

```
gap> IsPerm((1,2)(3,4));  
true  
gap> (1,2)(3,4)(5)=(1,2)(3,4);  
true  
gap> (1,2)(3,4)=(3,4)(2,1);  
true  
gap> IsPerm(25);  
false  
gap> IsPerm([1,2,3,4]);  
false
```

Permutations

The image of an element i under the natural right action of a permutation p is i^p . The preimage of the element i under p can be obtained with i/p . In the following example we compute the image of 1 and the preimage of 3 by the permutation (123):

```
gap> 2^(1,2,3);
```

```
3
```

```
gap> 2/(1,2,3);
```

```
1
```

Permutations

Composition of permutations will be performed from left to right.
For example

$$(123)(234) = (13)(24)$$

as the following code shows:

```
gap> (1,2,3) * (2,3,4);  
(1,3)(2,4)
```

To obtain the inverse of a permutation one uses `Inverse`:

```
gap> Inverse((1,2,3));  
(1,3,2)  
gap> (1,2,3)^(-1);  
(1,3,2)
```

Permutations

Let σ be a permutation written as a product of disjoint cycles. The function `ListPerm` returns a list containing $\sigma(i)$ at position i .

```
gap> # The permutation (12) in two letters
gap> ListPerm((1,2));
[ 2, 1 ]
gap> # The permutation (12) in four letters
gap> ListPerm((1,2), 4);
[ 2, 1, 3, 4 ]
gap> ListPerm((1,2,3)(4,5));
[ 2, 3, 1, 5, 4 ]
gap> ListPerm((1,3));
[ 3, 2, 1 ]
```

Conversely, any list of this type can be transformed into a permutation with the function `PermList`.

```
gap> PermList([1,2,3]);  
()  
gap> PermList([2,1]);  
(1,2)
```

Permutations

The **sign** of a permutation σ is the number $(-1)^k$, where one writes $\sigma = \tau_1 \cdots \tau_k$ as a product of transpositions. To compute the sign of a permutation one uses the function `SignPerm`.

```
gap> SignPerm(());  
1  
gap> SignPerm((1,2));  
-1  
gap> SignPerm((1,2,3,4,5));  
1  
gap> SignPerm((1,2)(3,4,5));  
-1  
gap> SignPerm((1,2)(3,4));  
1
```

An exercise on permutations

For a given positive integer n construct the permutation $\sigma \in \text{Sym}_n$ given by $\sigma(j) = n - j + 1$, Write σ as a product of disjoint cycles and compute its sign.

Finite fields

To create the finite field of p^n elements (here p is a prime number) we use the function `GF`. The **characteristic** of a field can be obtained with `Characteristic`.

```
gap> GF(2);  
GF(2)  
gap> GF(9);  
GF(3^2)  
gap> Characteristic(Rationals);  
0  
gap> Characteristic(CF(3));  
0  
gap> Characteristic(CF(4));  
0  
gap> Characteristic(GF(2));  
2  
gap> Characteristic(GF(9));  
3
```

Finite fields

Let p be a prime number and let F denote the field with $q = p^n$ elements, for some $n \in \mathbb{N}$. The subset

$$\{x \in F : x \neq 0\}$$

is a cyclic group of size $q - 1$; say generated by ζ . Then

$$F = \{0, \zeta, \zeta^2, \dots, \zeta^{q-1}\},$$

so each non-zero element of F is then a power of ζ .

```
gap> Size(GF(4));  
4  
gap> Elements(GF(4));  
[ 0*Z(2), Z(2)^0, Z(2^2), Z(2^2)^2 ]  
gap> Z(4);  
Z(2^2)  
gap> Inverse(Z(4));  
Z(2^2)^2
```

Finite fields

In GAP each non-zero element of the finite field $\text{GF}(q)$ will be a power of the generator $Z(q)$. The **zero** of $\text{GF}(q)$ will be $0 * Z(q)$ or equivalently $\text{Zero}(\text{GF}(q))$. $\text{One}(\text{GF}(q))$ will be the **multiplicative neutral element** of $\text{GF}(q)$.

```
gap> Zero(GF(4));  
0 * Z(2)  
gap> 0 in GF(4);  
false  
gap> Zero(Rationals);  
0  
gap> One(GF(4));  
Z(2)^0  
gap> 1 in GF(4);  
false  
gap> One(Rationals);  
1
```

Finite fields

To recognize elements in finite fields with a prime number of elements one uses the function `Int`.

```
gap> Elements(GF(5));  
[ 0*Z(5), Z(5)^0, Z(5), Z(5)^2, Z(5)^3 ]  
gap> Int(Z(5)^0);  
1  
gap> Int(Z(5)^1);  
2  
gap> Int(Z(5)^2);  
4  
gap> Int(Z(5)^3);  
3
```

An exercise on permutation polynomials

Prove that the map

$$f : \mathbb{Z}/8 \rightarrow \mathbb{Z}/8, \quad f(x) = 2x^2 + x,$$

defines a permutation on the ring $\mathbb{Z}/8$. Can you write this permutation as a product of disjoint cycles?

Matrices

For us a **matrix** will be just a rectangular array of numbers. The size of a matrix can be obtained with `DimensionsMat`. Sometimes (for example if one has an integer matrix) the function `Display` shows matrices in a nice way.

```
gap> m := [[1,2,3],[4,5,6]];;
gap> Display(m);
[ [ 1, 2, 3 ],
  [ 4, 5, 6 ] ]
gap> m[1][1];
1
gap> m[1][2];
2
gap> m[2][1];
4
gap> DimensionsMat(m);
[ 2, 3 ]
```

Matrices

Let $v = (1, 2, 3)$ and $w = (0, 5, -7)$ be row vectors of \mathbb{Q}^3 . Let us check that $-5v = (-5, -10, -15)$ and $2v - w = (2, -1, 13)$.

```
gap> v := [1,2,3];;  
gap> w := [0,5,-7];;  
gap> IsRowVector(v);  
true  
gap> IsRowVector(w);  
true  
gap> -5*v;  
[ -5, -10, -15 ]  
gap> 2*v-w;  
[ 2, -1, 13 ]
```

We also check that the **inner product** between v and w is -11 :

```
gap> v*w;  
-11
```

A very simple exercise with matrices

Let

$$A = \begin{pmatrix} 1 & 1 & 1 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix}, \quad B = \begin{pmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \end{pmatrix}, \quad C = \begin{pmatrix} 1 & 2 \\ 6 & 1 \\ 0 & 2 \end{pmatrix}.$$

Compute A^3 , BC , CB , $A + CB$ and $2A - 5CB$.

Matrices

To construct a **null matrix** one uses the function `NullMat`. The **identity** is constructed with the function `IdentityMat`. To construct **diagonal matrices** one uses `DiagonalMat`.

```
gap> Display(NullMat(2,3));  
[ [ 0, 0, 0 ],  
  [ 0, 0, 0 ] ]  
gap> Display(IdentityMat(3));  
[ [ 1, 0, 0 ],  
  [ 0, 1, 0 ],  
  [ 0, 0, 1 ] ]  
gap> Display(DiagonalMat([1,2]));  
[ [ 1, 0 ],  
  [ 0, 2 ] ]
```

Matrices

We know that `matrix[i][j]` returns the (i,j) -element of our matrix.
To extract **submatrices** from a matrix one uses

```
matrix{rows}{columns}
```

like in the following example:

```
gap> m := [\
> [1, 2, 3, 4, 5],\
> [6, 7, 8, 9, 3],\
> [3, 2, 1, 2, 4],\
> [7, 5, 3, 0, 0],\
> [0, 0, 0, 0, 1]];
gap> m{[1..3]}{[1..3]};
[ [ 1, 2, 3 ], [ 6, 7, 8 ], [ 3, 2, 1 ] ]
gap> m{[2,4,5]}{[1,3]};
[ [ 6, 8 ], [ 7, 3 ], [ 0, 0 ] ]
```

Matrices

It is possible to work with **matrices with coefficients in arbitrary rings**. Let us start working with matrices over the finite field of five elements:

```
gap> m := [[1,2,3],[3,2,1],[0,0,2]]*One(GF(5));  
[ [ Z(5)^0, Z(5), Z(5)^3 ],  
  [ Z(5)^3, Z(5), Z(5)^0 ],  
  [ 0*Z(5), 0*Z(5), Z(5) ] ]  
gap> Display(m);  
1 2 3  
3 2 1  
. . 2
```

Now let us work with 3×3 matrices with coefficients in the ring $\mathbb{Z}/4$. Let us compute the identity of $M_3(\mathbb{Z}/4)$:

```
gap> m := IdentityMat(3, ZmodnZ(4));;  
gap> Display(m);  
matrix over Integers mod 4:  
[ [ 1, 0, 0 ],  
  [ 0, 1, 0 ],  
  [ 0, 0, 1 ] ]
```

Matrices

One uses the function `Inverse` to compute the **inverse** of an invertible (square) matrix. This function returns `fail` if the matrix is not invertible.

```
gap> m := [[1, -2, -1], [0, 1, 0], [1, -1, 0]];;
gap> Display(Inverse(m));
[ [ 0, 1, 1 ],
  [ 0, 1, 0 ],
  [ -1, -1, 1 ] ]
gap> Inverse([[1,0],[2,0]]);
fail
```

`IsIdentityMat` returns either `true` if the argument is the identity matrix or `false` otherwise.

```
gap> IsIdentityMat(m*Inverse(m));
true
```

We use `TransposedMat` to compute the **transpose** of a matrix:

```
gap> m := [[1, -2, -1], [0, 1, 0], [1, -1, 0]];;  
gap> Display(TransposedMat(m)*m);  
[ [ 2, -3, -1 ],  
  [ -3, 6, 2 ],  
  [ -1, 2, 1 ] ]
```

A tricky exercise with matrices

Compute the eigenvalues and eigenvectors of the matrix

$$A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 6 & 7 & 8 \end{pmatrix} \in \mathbb{Q}^{3 \times 3}.$$

WARNING:

The function `Eigenvectors` returns generators of the eigenspaces, where $v \neq 0$ is an eigenvector of A with eigenvalue λ if and only if $vA = \lambda v$.